

CCIP 1.6 Documentation

[Glossary](#)

[Background](#)

[Goals](#)

[Scope:](#)

[OnRamp](#)

[Multi lane support](#)

[Nonce management](#)

[Rate limiting](#)

[Billing](#)

[OffRamp](#)

[Multi lane support](#)

[Nonce management](#)

[CommitStore merge](#)

[MultiOCR3](#)

[Batch commits](#)

[Batch executions](#)

[RMN](#)

[RMNRemote](#)

[RMNHome](#)

[FeeQuoter](#)

[Multi lane support](#)

[Keystone Price Reports](#)

[CCIPHome](#)

[CapabilityRegistry](#)

[CCIPHome](#)

[Blue/green](#)

[Non EVM Chains](#)

[FeeQuoter](#)

[OffRamp](#)

[Call Flow](#)

[Special Areas of Concern](#)

[Known / Out of Scope Issues:](#)

[Token developer is malicious](#)

[Token developer responsibility](#)

[Compatibility](#)

[Other](#)

Glossary

- **Blessing:** An explicit approval from an RMN node on a given commitment
- **Curse:** A warning - usually about a chain, but could be any arbitrary subject - that something is broken and shouldn't be used, e.g. a specific chain stopped producing new blocks
- **DON:** A *decentralized oracle network*, the set of Chainlink nodes coordinating to perform a specific action
- **Home chain:** The chain that `RMNHome` and `CCIPHome` contracts are deployed to
- **Job / Job Spec:** a configuration file that each Chainlink node uses to configure its participation in the DON running CCIP
- **Lane:** A message channel between two chains, ex Arb \Leftrightarrow OP and Arb \Leftrightarrow Polygon
- **NOP / Node Operator:** the person or organization behind a single node in a DON
- **OCR:** consensus protocol designed specifically for Chainlink nodes
- **Report:** the payload produced by OCR, signed by DON participants, and relayed to chain; there are two types of reports in CCIP: commit reports and exec (execute) reports
- **RMN:** the *risk management network*, a separate conglomerate of nodes specifically dedicated to observing and validating commit roots in CCIP

Background

For the most comprehensive background on Chainlink CCIP, see our public docs:

[Chainlink Documentation](#) | [Chainlink Documentation](#)

In prior CCIP versions (1.5 and below), a lane is established by a uni-directional connection between two sets of smart contracts residing on different chains. Each connecting set of contracts is considered to be a lane and each unidirectional lane operates with: 1 contract on the source chain (`EVM2EVMOnRamp`) and 2 contracts on the destination chain (`CommitStore` + `EVM2EVMOffRamp`).

For example, 2 users looking to transfer tokens from the same chain to 2 different chains will use different lanes and therefore 2 different sets of contracts:

- User 1 sending tokens from Ethereum \rightarrow Polygon will use:
 - `EVM2EVMOnRamp` on Ethereum
 - `EVM2EVMOffRamp` on Polygon
 - `CommitStore` on Polygon
- User 2 sending tokens from Ethereum \rightarrow Arbitrum will use:
 - `EVM2EVMOnRamp` on Ethereum
 - `EVM2EVMOffRamp` on Arbitrum
 - `CommitStore` on Arbitrum

Also, in 1.5 and below, each Risk Management Network node calls the Risk Management contract to “bless” the committed Merkle root. When there are enough blessing votes, the root becomes available for execution. In case of anomalies, each Risk Management node calls the Risk Management contract to “curse” the system. If the cursed quorum is reached, the Risk Management contract is paused to prevent any CCIP transaction from being executed.

Finally, prior CCIP versions have EVM specific logic hence the `EVM2EVMOn/OffRamp` names, making them incompatible with non EVM chains.

Goals


The main goal of CCIP v1.6 is to address the current limitations with regards to scalability by moving from a single on/off ramp per lane, to a single on/off ramp per chain, shared across lanes and chain families (sometimes referred to as a multi-lane architecture). By making the contracts non lane specific in v1.6, there are no more contracts to be deployed for new unidirectional lane, only configs must be set.

Another goal is to convert CCIP v1.6 contracts to be chain family agnostic (i.e. not tied to EVM), enabling bridging across different chain families. The major identified differences between chains are the below:

- Addresses may be different than 20 bytes.
- The resources used in transaction execution may be billed differently across families (for example Aptos and ZKsync charge for state changes separately).

There is still a desire to perform destination chain-specific validations when users send messages, there must be some family specific code on-chain. Therefore, a new chain family may require breaking changes, but the goal is to minimize it.

For high level contract interactions please refer to below diagram:



Scope:

The scope of this audit is confined to the core CCIP contracts outlined below, *token pool contracts are out of scope*.

OnRamp

The CCIP v1.6 `OnRamp` has been changed in the following ways:

1. Multiple lanes support: all lane specific variables can now be configured directly either on the `OnRamp` or the `FeeQuoter` instead of having to deploy a new contract
2. Chain family agnostic logic: prior ramp contracts were EVM specific, v1.6 contracts have been designed to be compatible with many chain families.
3. Outsourced fee and chain family specific logic to the `FeeQuoter`: this change makes the `OnRamp` destination chain family agnostic - future chain families would typically require only upgrading the `FeeQuoter` contract.
4. Outsourced nonce management logic: sender's nonce management are now managed by the `NonceManager` contract. This makes nonce preservation between upgrades significantly more efficient.
5. Outsourced rate limiting logic: rate limiting is now optional and managed by a hook `IMessageInterceptor` contract. This interceptor could do things besides rate limiting as well.
6. Outsourced billing logic: node operators payment logic has been outsourced to a separate `FeeAggregator` contract (not in scope of this audit).

Multi lane support

In order to support multiple lanes, some state variables have been updated. Destination chain specific variables are now stored in a `DestChainConfig` struct. Destination chain specific pricing configs have been moved to the `FeeQuoter` (previously named `PriceRegistry`).

```
1  /// @dev Struct to hold the configs for a destination chain
2  /// @dev sequenceNumber, allowListEnabled, router will all be packed in 1 slot
3  struct DestChainConfig {
4      // The last used sequence number. This is zero in the case where no messages have yet been sent.
5      // 0 is not a valid sequence number for any real transaction.
6      uint64 sequenceNumber; // ── The last used sequence number
7      bool allowListEnabled; // | boolean indicator to specify if allowList check is enabled
8      IRouter router; // ───────── Local router address that is allowed to send messages to the destination chain.
9      EnumerableSet.AddressSet allowedSendersList; // The list of addresses allowed to send messages from OnRamp
10 }
```

Nonce management

The new release also introduces a `NonceManager` to extract the sender nonce management logic from both the `OnRamp` and `OffRamp`. This brings multiple benefits including:

- Contract size reduction for the `OnRamp` and the `OffRamp`.
- Makes nonce reasoning during `OffRamp` upgrades a lot easier, given there is a single source of truth for nonces across all versions of a lane.
- `prevOnRamp` & `prevOffRamp` only need to be set once.

At the cost of:

- Extra gas cost, but reduced gas costs compared to long `prevOnRamp` or `prevOffRamp` chains previously possible.
- Extra config management (`NonceManager` permissions).
- Adding a contract to the deployment pipeline for new chains.

The `NonceManager` `getIncrementedOutboundNonce()` function is now called from the `OnRamp` `forwardFromRouter` function (if out of order execution (OOO) is disabled):

```
1 newMessage.header.nonce = isOutOfOrderExecution
2   ? 0
3   : INonceManager(i_nonceManager).getIncrementedOutboundNonce(destChainSelector, originalSender);
4 newMessage.extraArgs = convertedExtraArgs;
```

And the `NonceManager` will perform the nonce bump and potential callback to the previous onramp to ensure nonce sequencing:

```
1 function getIncrementedOutboundNonce (
2     uint64 destChainSelector,
3     address sender
4 ) external onlyAuthorizedCallers returns (uint64) {
5     uint64 outboundNonce = _getOutboundNonce(destChainSelector, sender) + 1;
6     s_outboundNonces[destChainSelector][sender] = outboundNonce;
7
8     return outboundNonce;
9 }
10
11 function _getOutboundNonce(uint64 destChainSelector, address sender) private view returns (uint64) {
12     uint64 outboundNonce = s_outboundNonces[destChainSelector][sender];
13
14     // When introducing the NonceManager with existing lanes, we still want to have sequential nonces.
15     // Referencing the old onRamp preserves sequencing between updates.
16     if (outboundNonce == 0) {
17         address prevOnRamp = s_previousRamps[destChainSelector].prevOnRamp;
18         if (prevOnRamp != address(0)) {
19             return IEVM2AnyOnRamp(prevOnRamp).getSenderNonce(sender);
20         }
21     }
22
23     return outboundNonce;
24 }
```

Rate limiting

In the existing deployment, the `OnRamp` implements an `AggregateRateLimiter` (ARL), in v1.6 the decision was made to remove the ARL from the `onRamp` and `offRamp` to save contract size and allow us to do various other checks without having to deploy new ramps. The new approach is to configure an optional message validator and separately deploy the rate limiter validation contract.

```
1 address messageInterceptor = s_dynamicConfig.messageInterceptor;
2 if (messageInterceptor != address(0)) {
3     IMessageInterceptor(messageInterceptor).onOutboundMessage(destChainSelector, message);
4 }
```

The only requirement of the message validator contract is that it implements the `IMessageInterceptor` interface:

```
1 /// @notice Interface for plug-in message hook contracts that intercept OffRamp & OnRamp messages
2 /// and perform validations / state changes on top of the messages. The interceptor functions are expected to
3 /// revert on validation failures.
4 interface IMessageInterceptor {
5     /// @notice Common error that can be thrown on validation failures and used by consumers
6     /// @param errorReason abi encoded revert reason
7     error MessageValidationError(bytes errorReason);
8
9     /// @notice Intercepts & validates the given OffRamp message. Reverts on validation failure
10    /// @param message to validate
11    function onInboundMessage(Client.Any2EVMMessage memory message) external;
12
13    /// @notice Intercepts & validates the given OnRamp message. Reverts on validation failure
14    /// @param destChainSelector remote destination chain selector where the message is being sent to
15    /// @param message to validate
16    function onOutboundMessage(uint64 destChainSelector, Client.EVM2AnyMessage memory message) external;
17 }
```

So it does not have to be a rate limiter contract, but as of now, the only implementation is the aggregate rate limiter. The main difference between the previous `AggregateRateLimiter` and the v1.6 `MultiAggregateRateLimiter` is that, similarly to the `OnRamp`, the contract supports multiple chain rate limitation (with separate buckets for the `OnRamp` and `OffRamp`).

```
1 /// @notice Struct to store rate limit token buckets for both lane directions
2 struct RateLimiterBuckets {
3     RateLimiter.TokenBucket inboundLaneBucket; // Bucket for the inbound lane (remote -> local)
```

```

4   RateLimiter.TokenBucket outboundLaneBucket; // Bucket for the outbound lane (local -> remote)
5 }
6
7 /// @notice Rate limiter token bucket states per chain, with separate buckets for incoming and outgoing lanes.
8 mapping(uint64 remoteChainSelector => RateLimiterBuckets buckets) s_rateLimitersByChainSelector;

```

Billing

This release extracts Node Operator (NOP) payment logic to a feeAggregator contract (not in scope).

The `OnRamp` `s_nopFeesJuels` state variable has been replaced by adding `feeValueJuels` to `CCIPMessageSent`, the CCIP message event. `feeValueJuels` is the fee amount in the smallest Link token denomination: `juels` (equal to `wei` for Ethereum). `CCIPMessageSent` will be used off-chain to calculate the total amount owed to NOPs.

```

1 (newMessage.feeValueJuels, isOutOfOrderExecution, convertedExtraArgs, destExecDataPerToken) = IFeeQuoter(
2   s_dynamicConfig.feeQuoter
3 ).processMessageArgs(
4   destChainSelector, message.feeToken, feeTokenAmount, message.extraArgs, newMessage.tokenAmounts, tokenAmounts
5 );

```

OffRamp

The `OffRamp` has been changes in the follow ways:

1. Multiple lanes support: multiple source chain specific variables can now be configured directly on the `OffRamp` instead of having to deploy a new contract.
2. Chain family agnostic logic: existing ramp contracts were EVM specific contracts, v1.6 contracts have been designed to be compatible with many chain families.
3. Outsourced nonce management logic: sender's nonce management are now managed by a `NonceManager` contract
4. `CommitStore` merge: The `CommitStore` has been merged into the `OffRamp` to reduce the number of contracts to deploy and maintain. This also has some gas and safety benefits.
5. `RMN` off-chain blessing

Multi lane support

In order to support multiple lanes, some state variables have been updated. All source chain specific variables are now stored in a `SourceChainConfig` struct with the exception of execution states and committed roots which require a separate nested mapping.

Nonce management

As mentioned in the `OnRamp` nonce management section, the `OffRamp` also outsources its nonce management logic to the `NonceManager`. It does so by calling the `NonceManager` `incrementInboundNonce()` function which conditionally bumps the nonce. The nonce is only incremented if the next nonce matches the expected nonce provided by the `OffRamp` message.

If the nonce hasn't been incremented then the `OffRamp` skips the message execution:

```

1 if (message.header.nonce != 0) {
2   if (originalState == Internal.MessageExecutionState.UNTOUCHED) {
3     // If a nonce is not incremented, that means it was skipped, and we can ignore the message
4     if (
5       !INonManager(i_nonceManager).incrementInboundNonce(
6         sourceChainSelector, message.header.nonce, message.sender
7       )
8     ) continue;
9   }
10 }

```

CommitStore merge

The `CommitStore` and `OffRamp` are now a single contract. This merge adds the following functionalities:

- Multiple OCR configurations
- Batch commits
- Batch executions

MultiOCR3

Since the `OffRamp` contract now needs to support both commit and execution reports, multiple OCR configurations are required. Hence, a new `MultiOCR3` contract was introduced.

The `OCRConfigArgs` struct (used to set new OCR configs) now has a new `ocrPluginType` field, which in the case of the `OffRamp` will help differentiate between the commit and execution plugin OCR config:

```

1 /// @notice OCR configuration for a single OCR plugin within a DON.
2 struct OCRConfig {
3   ConfigInfo configInfo; // latest OCR config
4   // NOTE: len(signers) can be different from len(transmitters). There is no index relationship between the two arrays
5   address[] signers; // addresses oracles use to sign the reports
6   address[] transmitters; // addresses oracles use to transmit the reports
7 }
8
9 /// @notice Args to update an OCR Config.
10 struct OCRConfigArgs {
11   bytes32 configDigest; // Config digest to update to
12   uint8 ocrPluginType; // _____, OCR plugin type to update config for
13   uint8 F; // | maximum number of faulty/dishonest oracles
14   bool isSignatureVerificationEnabled; // —/ if true, requires signers and verifies signatures on transmission verification
15   address[] signers; // signing address of each oracle
16   address[] transmitters; // transmission address of each oracle (i.e. the address the oracle actually sends transactions to the contract from)
17 }

```

And both OCR configs and oracles are now tracked in a mapping with the `ocrPulginType` as its key:

```
1 /// @notice mapping of OCR plugin type -> DON config
2 mapping(uint8 ocrPluginType => OCRConfig config) internal s_ocrConfigs;
3
4 /// @notice OCR plugin type => signer OR transmitter address mapping
5 mapping(uint8 ocrPluginType => mapping(address signerOrTransmitter => Oracle oracle)) internal s_oracles;
```

The `_transmit` function now also takes in an `ocrPluginType` and is called directly by either the `commit()` or the `execute()` function of the `OffRamp`.

Batch commits

As mentioned in the scalability section the lack of cross-source message batching has been one of the identified drawbacks of v1.5. In v1.6, the `OffRamp` now takes in a array of merkle roots:

```
1 /// @dev Struct to hold a merkle root and an interval for a source chain so that an array of these can be passed in the CommitReport.
2 /// @dev RMN depends on this struct, if changing, please notify the RMN maintainers.
3 /// @dev inefficient struct packing intentionally chosen to maintain order of specificity. Not a storage struct so impact is minimal.
4 // solhint-disable-next-line gas-struct-packing
5 struct MerkleRoot {
6     uint64 sourceChainSelector; // Remote source chain selector that the Merkle Root is scoped to
7     bytes onRampAddress; // Generic onramp address, to support arbitrary sources; for EVM, use abi.encode
8     uint64 minSeqNr; // Minimum sequence number, inclusive
9     uint64 maxSeqNr; // Maximum sequence number, inclusive
10    bytes32 merkleRoot; // Merkle root covering the interval & source chain messages
11 }
12
13 /// @dev Report that is committed by the observing DON at the committing phase
14 /// @dev RMN depends on this struct, if changing, please notify the RMN maintainers.
15 struct CommitReport {
16     Internal.PriceUpdates priceUpdates; // Collection of gas and price updates to commit
17     Internal.MerkleRoot[] merkleRoots; // Collection of merkle roots per source chain to commit
18     IRMNv2.Signature[] rmnSignatures; // RMN signatures on the merkle roots
19 }
```

This enables transmitting multiple roots from multiple chains:

```
1 for (uint256 i = 0; i < commitReport.merkleRoots.length; ++i) {
2     Internal.MerkleRoot memory root = commitReport.merkleRoots[i];
3     uint64 sourceChainSelector = root.sourceChainSelector;
4
5     if (i_rmn.isCursed(bytes16(uint128(sourceChainSelector)))) {
6         revert CursedByRMN(sourceChainSelector);
7     }
8
9     SourceChainConfig storage sourceChainConfig = _getEnabledSourceChainConfig(sourceChainSelector);
10
11     if (keccak256(root.onRampAddress) != keccak256(sourceChainConfig.onRamp)) {
12         revert CommitOnRampMismatch(root.onRampAddress, sourceChainConfig.onRamp);
13     }
14
15     if (sourceChainConfig.minSeqNr != root.minSeqNr || root.minSeqNr > root.maxSeqNr) {
16         revert InvalidInterval(root.sourceChainSelector, root.minSeqNr, root.maxSeqNr);
17     }
18
19     bytes32 merkleRoot = root.merkleRoot;
20     if (merkleRoot == bytes32(0)) revert InvalidRoot();
21     // If we reached this section, the report should contain a valid root
22     // We disallow duplicate roots as that would reset the timestamp and
23     // delay potential manual execution.
24     if (s_roots[root.sourceChainSelector][merkleRoot] != 0) {
25         revert RootAlreadyCommitted(root.sourceChainSelector, merkleRoot);
26     }
27
28     sourceChainConfig.minSeqNr = root.maxSeqNr + 1;
29     s_roots[root.sourceChainSelector][merkleRoot] = block.timestamp;
30 }
```

Batch executions

Similarly to batch commits, the v1.6 `OffRamp` also enables batch executions for multiple reports from different source chains through the `execute()` function which takes in an array of `ExecutionReportSingleChain`:

```
1 /// @notice Report that is submitted by the execution DON at the execution phase. (including chain selector data)
2 /// @dev RMN depends on this struct, if changing, please notify the RMN maintainers.
3 struct ExecutionReportSingleChain {
4     uint64 sourceChainSelector; // Source chain selector for which the report is submitted
5     Any2EVMRampMessage[] messages;
6     // Contains a bytes array for each message, each inner bytes array contains bytes per transferred token
7     bytes[][] offchainTokenData;
8     bytes32[] proofs;
9     uint256 proofFlagBits;
10 }
11
12 /// @notice Transmit function for execution reports. The function takes no signatures,
```

```

13 /// and expects the exec plugin type to be configured with no signatures.
14 /// @param report serialized execution report
15 function execute(bytes32[3] calldata reportContext, bytes calldata report) external {
16     _batchExecute(abi.decode(report, (Internal.ExecutionReportSingleChain[])), new GasLimitOverride[[]](0));
17
18     bytes32[] memory emptySigs = new bytes32[](0);
19     _transmit(uint8(Internal.OCRPluginType.Execution), reportContext, report, emptySigs, emptySigs, bytes32(""));
20 }

```

RMN

RMNRemote

The fundamental RMN change in CCIP v1.6 is to move RMN blessings from on-chain to off-chain. In prior versions, RMN nodes each sent transactions to chain to explicitly bless a root *after* it was committed, and until that root was blessed, no messages contained within could be executed. In 1.6, blessings are collected *offchain* and included in the commit report. Blessings are then verified on chain as part of the `commit()` transaction.

Each `OffRamp` contains a pointer to a `RMNRemote` contract. The `RMNRemote` contract exposes a `verify()` function:

```

1 interface IRMNRemote {
2     /// @notice Verifies signatures of RMN nodes, on dest lane updates as provided in the CommitReport
3     /// @param offRampAddress is not inferred by msg.sender, in case the call is made through ARMPProxy
4     /// @param merkleRoots must be well formed, and is a representation of the CommitReport received from the oracles
5     /// @param signatures rmnNodes ECDSA sigs, only r & s, must be sorted in ascending order by signer address
6     /// @param rawVs rmnNodes ECDSA sigs, part v bitmap
7     /// @dev Will revert if verification fails
8     function verify(
9         address offRampAddress,
10        Internal.MerkleRoot[] memory merkleRoots,
11        Signature[] memory signatures,
12        uint256 rawVs
13    ) external view;
14 }

```

This `verify()` function is used by the `OffRamp` to validate merkle roots before persisting them:

```

1 // OffRamp.sol
2
3 function commit(...) {
4     // ...
5     if (commitReport.merkleRoots.length > 0) {
6         i_rmnRemote.verify(address(this), commitReport.merkleRoots, commitReport.rmnSignatures, commitReport.rmnRawVs);
7     }
8     // ...
9 }
10
11

```

Note that RMN signatures are only validated *if* this report contains roots to commit (it might only have price updates).

The `RMNRemote` contract also has the ability to *curse* arbitrary subjects

```

1 function curse(bytes16 subject) external;

```

In 1.5, cursing was done automatically by RMN nodes when they detected anomalies. In 1.6, only manual curses are possible, where the RMN contract owner explicitly curses a subject.

RMNHome

The `RMNHome` contract exists only to configure and coordinate offchain systems. It has no direct interactions with other contracts. The `RMNHome` contract contains a `VersionedConfig` struct that is consumed by the RMN and CommitDON to ensure all commits and blessings are made based on the latest config. Ex the `RMNRemote` contract contains:

```

1 // RMNRemote.sol
2
3 struct Config {
4     bytes32 rmnHomeContractConfigDigest; // Digest of the RMNHome contract config
5     Signer[] signers;
6     uint64 minSigners;
7 }

```

Above, the `rmnHomeContractConfigDigest` is a digest of the RMNHome contract's config. When the `RMNHome` config *changes*, the `RMNRemote` contracts on each chain need to be updated to account for the new config. Any commit reports generated with mis-matched config digests will fail to persist on-chain. This helps ensure config changes are atomic and all commitments / blessings are generated on the latest config.

`RMNHome` can support two concurrent configurations - one "live" and one "pending". These are used as part of CCIP blue/green deployments.

FeeQuoter

Multi lane support

As mentioned in the [OnRamp](#) section, all fee and chain family specific logic have been transferred from the `OnRamp` to the `FeeQuoter` to make the `OnRamp` fully destination chain family agnostic. Therefore, similarly to the `OnRamp`, all destination chain specific variables are now stored in a `DestChainConfig` struct with the exception of the token transfer fee configs which require a separate nested mapping.

Variable		Per destination chain	Per token
<code>isEnabled</code>	Whether a destination chain is enabled	✓	
<code>maxNumberOfTokensPerMsg</code>	Maximum number of distinct ERC20 token transferred per message	✓	
<code>maxDataBytes</code>	Maximum payload data size in bytes	✓	
<code>maxPerMsgGasLimit</code>	Maximum gas limit for messages targeting EVMs	✓	
<code>destGasOverhead</code>	Gas charged on top of the gasLimit to cover destination chain costs	✓	
<code>destGasPerPayloadByte</code>	Destination chain gas charged for passing each byte of data payload to receiver	✓	
<code>destDataAvailabilityOverheadGas</code>	Extra data availability gas charged on top of the message, e.g. for OCR	✓	
<code>destGasPerDataAvailabilityByte</code>	Amount of gas to charge per byte of message data that needs availability	✓	
<code>destDataAvailabilityMultiplierBps</code>	Multiplier for data availability gas, multiples of bps	✓	
<code>defaultTokenFeeUSDCents</code>	Default token fee charged per token transfer	✓	
<code>defaultTokenDestGasOverhead</code>	Default gas charged to execute the token transfer on the destination chain	✓	
<code>defaultTokenDestBytesOverhead</code>	Default extra data availability bytes charged per token transfer	✓	
<code>defaultTxGasLimit</code>	Default gas limit for a tx	✓	
<code>gasMultiplierWeiPerEth</code>	Multiplier for gas costs, 1e18 based so 11e17 = 10% extra cost	✓	
<code>networkFeeUSDCents</code>	Flat network fee to charge for messages, multiples of 0.01 USD	✓	
<code>gasPriceStalenessThreshold</code>	The amount of time a gas price can be stale before it is considered invalid (0 means disabled)	✓	
<code>enforceOutOfOrder</code>	Whether to enforce the allowOutOfOrderExecution extraArg value to be true	✓	
<code>chainFamilySelector</code>	Selector that identifies the destination chain's family. Used to determine the correct validations to perform for the dest chain	✓	
<code>s_usdPerUnitGasByDestChainSelector</code>	The gas price per unit of gas for a given destination chain, in USD with 18 decimals	✓	
<code>minFeeUSDCents</code>	Minimum fee to charge per token transfer, multiples of 0.01 USD	✓	✓
<code>maxFeeUSDCents</code>	Maximum fee to charge per token transfer, multiples of 0.01 USD	✓	✓
<code>deciBps</code>	Basis points charged on token transfers, multiples of 0.1bps, or 1e-5	✓	✓
<code>destGasOverhead</code>	Gas charged to execute the token transfer on the destination chain	✓	✓
<code>destBytesOverhead</code>	Default data availability bytes that are returned from the source pool and sent to the destination pool	✓	✓
<code>s_usdPerToken</code>	The price, in USD with 18 decimals, per 1e18 of the smallest token denomination		✓
<code>s_usdPriceFeedsPerToken</code>	Stores the price data feed configurations per token.		✓
<code>s_premiumMultiplierWeiPerEth</code>	The multiplier for destination chain specific premiums that can be set by the owner or fee admin. This should never be 0 once set, so it can be used as an isEnabled flag		✓

These variables can be updated by the contract owner through the `applyDestChainConfigUpdates()` function:

```

1 /// @notice Updates destination chains specific configs.
2 /// @param destChainConfigArgs Array of destination chain specific configs.
```

```

3 function applyDestChainConfigUpdates(DestChainConfigArgs[] memory destChainConfigArgs) external onlyOwner {
4     _applyDestChainConfigUpdates(destChainConfigArgs);
5 }
6
7 /// @notice Internal version of applyDestChainConfigUpdates.
8 function _applyDestChainConfigUpdates(DestChainConfigArgs[] memory destChainConfigArgs) internal {
9     for (uint256 i = 0; i < destChainConfigArgs.length; ++i) {
10         DestChainConfigArgs memory destChainConfigArg = destChainConfigArgs[i];
11         uint64 destChainSelector = destChainConfigArgs[i].destChainSelector;
12
13         if (destChainSelector == 0) {
14             revert InvalidDestChainConfig(destChainSelector);
15         }
16
17         DestChainConfig storage destChainConfig = s_destChainConfigs[destChainSelector];
18         destChainConfig.router = destChainConfigArg.router;
19
20         emit DestChainConfigSet(
21             destChainSelector, destChainConfig.sequenceNumber, destChainConfigArg.router, destChainConfig.allowListEnabled
22         );
23     }
24 }

```

Keystone Price Reports

In CCIP v1.6 the `FeeQuoter` supports token prices transmissions from both the commit plugin and [Keystone*](#). Therefore, the `FeeQuoter` now conforms to the `IReceiver` interface and implements the `onReport()` function:

```

1 /// @notice Handles the report containing price feeds and updates the internal price storage
2 /// @inheritdoc IReceiver
3 /// @dev This function is called to process incoming price feed data.
4 /// @param metadata Arbitrary metadata associated with the report (not used in this implementation).
5 /// @param report Encoded report containing an array of 'ReceivedCCIPFeedReport' structs.
6 function onReport(bytes calldata metadata, bytes calldata report) external {
7     (bytes10 workflowName, address workflowOwner, bytes2 reportName) = metadata._extractMetadataInfo();
8
9     _validateReportPermission(msg.sender, workflowOwner, workflowName, reportName);
10
11     ReceivedCCIPFeedReport[] memory feeds = abi.decode(report, (ReceivedCCIPFeedReport[]));
12
13     for (uint256 i = 0; i < feeds.length; ++i) {
14         uint8 tokenDecimals = s_usdPriceFeedsPerToken[feeds[i].token].tokenDecimals;
15         if (tokenDecimals == 0) {
16             revert TokenNotSupported(feeds[i].token);
17         }
18         // Keystone reports prices in USD with 18 decimals, so we passing it as 18 in the _calculateRebasedValue function
19         uint224 rebasedValue = _calculateRebasedValue(18, tokenDecimals, feeds[i].price);
20
21         //if stale update then revert
22         if (feeds[i].timestamp < s_usdPerToken[feeds[i].token].timestamp) {
23             revert StaleKeystoneUpdate(feeds[i].token, feeds[i].timestamp, s_usdPerToken[feeds[i].token].timestamp);
24         }
25
26         s_usdPerToken[feeds[i].token] =
27             Internal.TimestampedPackedUint224({value: rebasedValue, timestamp: feeds[i].timestamp});
28         emit UsdPerTokenUpdated(feeds[i].token, rebasedValue, feeds[i].timestamp);
29     }
30 }

```

*Keystone is a new form of price reporting and the details of its implementation are not relevant this documentation.

CCIPHome

CapabilityRegistry

In CCIP v1.6, source and destination NOPs are decoupled. This is achieved through the use of a singular home chain contract on Ethereum mainnet called the `CapabilityRegistry` that contains configuration readable by all NOPs (the `CapabilityRegistry` is out of scope for this review). This does force NOPs to support the home chain but is a practical compromise, and it could, in principle, be moved to another non-chain location.

The `CapabilityRegistry` contains explicitly:

- All CCIP DONs and their configuration
- Each DON contains:
 - The list of participating nodes (in particular, their P2P IDs)
 - The capabilities supported by each node, including the capability configurations.

CCIPHome

Each capability can point to a configuration contract that contains capability specific config logic. In the case of CCIP, that contract is the `CCIPHome` contract. This contract is therefore, like the `CapabilityRegistry`, only deployed on the home chain (i.e. Ethereum) and must conform to the `ICapabilityConfiguration`, with 2 functions:

- `beforeCapabilityConfigSet` which is called by the `CapabilityRegistry` upon adding or updated a config for a particular DON through the `updateDON` function
- `getCapabilityConfiguration` which fetches a capability configuration for a particular DON instance

The role of `CCIPHome` is to store and manage the configuration for the CCIP capability.

There are 2 classes of configuration: the chain configuration and the DON configuration. Each chain has a single configuration:

```
1  /// @notice Chain configuration.
2  /// Changes to chain configuration are detected out-of-band in plugins and decoded offchain.
3  struct ChainConfig {
4      bytes32[] readers; // The P2P IDs of the readers for the chain. These IDs must be registered in the capabilities registry.
5      uint8 fChain; // The fault tolerance parameter of the chain.
6      bytes config; // The chain configuration. This is kept intentionally opaque so as to add fields in the future if needed.
7  }
8
9  /// @notice Chain configuration information struct used in applyChainConfigUpdates and getAllChainConfigs.
10 struct ChainConfigInfo {
11     uint64 chainSelector;
12     ChainConfig chainConfig;
13 }
```

And each DON will have up to 4 configurations, 1 for each plugin (commit and exec), one blue and one green:

```
1  /// @notice OCR3 configuration.
2  /// Note that FRoleDON >= fChain, since FRoleDON represents the role DON, and fChain represents sub-committees.
3  /// FRoleDON values are typically identical across multiple OCR3 configs since the chains pertain to one role DON,
4  /// but FRoleDON values can change across OCR3 configs to indicate role DON splits.
5  struct OCR3Config {
6      Internal.OCRPluginType pluginType; // ----- \ The plugin that the configuration is for.
7      uint64 chainSelector; // | The (remote) chain that the configuration is for.
8      uint8 FRoleDON; // | The "big F" parameter for the role DON.
9      uint64 offchainConfigVersion; // ----- / The version of the offchain configuration.
10     bytes offrampAddress; // The remote chain offramp address.
11     OCR3Node[] nodes; // Keys & IDs of nodes part of the role DON
12     bytes offchainConfig; // The offchain configuration for the OCR3 protocol. Protobuf encoded.
13 }
14
15 /// @notice OCR3 configuration with metadata, specifically the config count and the config digest.
16 struct OCR3ConfigWithMeta {
17     OCR3Config config; // The OCR3 configuration.
18     uint64 configCount; // The config count used to compute the config digest.
19     bytes32 configDigest; // The config digest of the OCR3 configuration.
20 }
```


Blue/green

In CCIPv1.6, there is the desire for NOPS to support precisely whichever chains they want to support, but still have interoperability between all chains by default. To accomplish that, there is one OCR instance per destination chain (`OCR3ConfigWithMeta`) with a heterogeneous set of members supporting different chains (including ones who cannot read/write to the destination chain)

All OCR configurations are held on the home chain (`CapabilityRegistry` + `CCIPHome`) which all CCIP NOPS can read from. This way they can join an OCR instance which writes to chains even if they don't support it. Remote chains need to validate signatures on the OCR reports produced, meaning signer changes necessarily require touching both chains.

When performing an upgrade, a blue/green deployment model is used, where 2 versions of CCIP are run in parallel until the upgrade is complete. The active version is referred to as the "blue" version, while the upgraded version is initially set as the "green" version. The "green" version acts as a staging environment to enable an upgrade without downtime and testing. After the upgrade is complete, the "green" (v2) version is promoted to the "blue" version, and the previous "blue" (v1) version is terminated.

Both instances use custom contract transmitters which don't submit to the remote until its digest is configured. It can therefore be rolled back if v2 instance looks unhealthy by setting just the v1 instance (blue) on the home chain. Once the v2 digest is set on the remote chain, it unblocks the v2 instance and disables the v1 instance. The v1 instance can then be disabled on the home chain.



1. Nodes perform `commit()` and `execute()` operations on the v1 `OffRamp`.
2. `updateDON()` is called on the `CapabilityRegistry` for each destination chain with the updated list of nodes and the corresponding blue and green configs.
3. `beforeCapabilityConfigSet()` is called on the `CCIPHome` contract, which adds the green config for each plugin.
4. Nodes pick up the `ConfigSet` event from the `CapabilityRegistry` and call `getOCRConfig()` on the `CCIPHome` contract to pull the green config + digest, which is used to map the v1.6 `OffRamp` of the destination chain.
5. `setOCR3Configs()` is called on the v2 `OffRamp`
6. The `ConfigSet` event is picked by the nodes
7. At this point both blue and green config are operational and nodes can perform `commit()` and `execute()` on both the v1 and v2 `OffRamp`
8. `updateDON` is called on the `CapabilityRegistry` for each destination chain with only the green configs
9. The green config now becomes the new blue config

Non EVM Chains

FeeQuoter

The following struct changes have been implemented for any chain support:

- A `familyTag` variable has been added to the `DestChainDynamicConfig` struct to identify the destination chain's family. It is used to determine the correct validations to perform for the dest chain. For EVM chains the used family tag can be found in the `Internal` library:

```

1 // bytes4(keccak256("CCIP ChainFamilySelector EVM"))
2 bytes4 public constant CHAIN_FAMILY_SELECTOR_EVM = 0x2812d52c;

```

- An `EVM2AnyRampMessage` struct has been introduced, replacing the `EVM2EVMMessage` struct for new v1.6 lanes, where all destination specific `address` type variables have been updated to `bytes`:

```

1 /// @notice Family-agnostic header for OnRamp & OffRamp messages.
2 /// The messageId is not expected to match hash(message), since it may originate from another ramp family
3 struct RampMessageHeader {
4     bytes32 messageId; // Unique identifier for the message, generated with the source chain's encoding scheme (i.e. not necessarily abi encoded)
5     uint64 sourceChainSelector; // the chain selector of the source chain, note: not chainId
6     uint64 destChainSelector; // the chain selector of the destination chain, note: not chainId
7     uint64 sequenceNumber; // sequence number, not unique across lanes
8     uint64 nonce; // nonce for this lane for this sender, not unique across senders/lanes
9 }
10
11 struct EVM2AnyTokenTransfer {
12     // The source pool EVM address. This value is trusted as it was obtained through the onRamp. It can be
13     // relied upon by the destination pool to validate the source pool.
14     address sourcePoolAddress;
15     // The EVM address of the destination token
16     // This value is UNTRUSTED as any pool owner can return whatever value they want.
17     bytes destTokenAddress;
18     // Optional pool data to be transferred to the destination chain. Be default this is capped at
19     // CCIP_LOCK_OR_BURN_V1_RET_BYTES bytes. If more data is required, the TokenTransferFeeConfig.destBytesOverhead
20     // has to be set for the specific token.
21     bytes extraData;
22     uint256 amount; // Amount of tokens.
23     // Destination chain specific execution data encoded in bytes

```

```

24 // for an EVM destination, it consists of the amount of gas available for the releaseOrMint
25 // and transfer calls made by the offRamp
26 bytes destExecData;
27 }
28
29 /// @notice Family-agnostic message emitted from the OnRamp
30 /// Note: hash(Any2EVMRampMessage) != hash(EVM2AnyRampMessage) due to encoding & parameter differences
31 /// messageId = hash(EVM2AnyRampMessage) using the source EVM chain's encoding format
32 struct EVM2AnyRampMessage {
33     RampMessageHeader header; // Message header
34     address sender; // sender address on the source chain
35     bytes data; // arbitrary data payload supplied by the message sender
36     bytes receiver; // receiver address on the destination chain
37     bytes extraArgs; // destination-chain specific extra args, such as the gasLimit for EVM chains
38     address feeToken; // fee token
39     uint256 feeTokenAmount; // fee token amount
40     uint256 feeValueJuels; // fee amount in Juels
41     EVM2AnyTokenTransfer[] tokenAmounts; // array of tokens and amounts to transfer
42 }

```

As mentioned in the [goals](#) section, there is still a desire to perform destination chain-specific validations, therefore the following logic changes have been implemented:

- Family specific address validation logic has been implemented: if the destination chain is an EVM chain then use the internal `_validateEVMAddress` function, otherwise new internal functions will likely be introduced to perform different sanity checks.

```

1 /// @notice Validates that the destAddress matches the expected format of the family.
2 /// @param chainFamilySelector Tag to identify the target family.
3 /// @param destAddress Dest address to validate.
4 /// @dev precondition - assumes the family tag is correct and validated.
5 function _validateDestFamilyAddress(bytes4 chainFamilySelector, bytes memory destAddress) internal pure {
6     if (chainFamilySelector == Internal.CHAIN_FAMILY_SELECTOR_EVM) {
7         Internal._validateEVMAddress(destAddress);
8     }
9 }
10
11 /// @notice This methods provides validation for parsing abi encoded addresses by ensuring the
12 /// address is within the EVM address space. If it isn't it will revert with an InvalidEVMAddress error, which
13 /// we can catch and handle more gracefully than a revert from abi.decode.
14 /// @return The address if it is valid, the function will revert otherwise.
15 function _validateEVMAddress(bytes memory encodedAddress) internal pure returns (address) {
16     if (encodedAddress.length != 32) revert InvalidEVMAddress(encodedAddress);
17     uint256 encodedAddressUint = abi.decode(encodedAddress, (uint256));
18     if (encodedAddressUint > type(uint160).max || encodedAddressUint < PRECOMPILE_SPACE) {
19         revert InvalidEVMAddress(encodedAddress);
20     }
21     return address(uint160(encodedAddressUint));
22 }

```

- Custom `extraArgs` parsing logic: since the `extraArgs` content might change depending on the destination chain, decoding must be customized. For example, a decoded Solana `extraArgs` might look something like:

```

1 struct SolExtraArgsV1 {
2     bool allowOutOfOrder; // Required true initially (possibly forever, seems)
3     uint64 computeUnits; // 200k default, 1.4M max https://solana.com/docs/core/fees#compute-unit-limit-1
4     uint64 computeUnitPrice; // Optional set gas price https://solana.com/docs/core/fees#prioritization-fees-1
5     SolTokenAccountMeta[] tokenAccountMetas;
6 }

```

To map the `extraArgs` to its chain family specific decoding the chain family selector is prepended to the encoded bytes. The parsing can then be performed in the `_parseEVMExtraArgsFromBytes()` function:

```

1 /// @dev Convert the extra args bytes into a struct with validations against the dest chain config.
2 /// @param extraArgs The extra args bytes.
3 /// @param destChainConfig Dest chain config to validate against.
4 /// @return evmExtraArgs The EVMExtraArgs struct (latest version).
5 function _parseEVMExtraArgsFromBytes(
6     bytes calldata extraArgs,
7     DestChainConfig memory destChainConfig
8 ) internal pure returns (Client.EVMExtraArgsV2 memory) {
9     Client.EVMExtraArgsV2 memory evmExtraArgs =
10         _parseUnvalidatedEVMExtraArgsFromBytes(extraArgs, destChainConfig.defaultTxGasLimit);
11
12     if (evmExtraArgs.gasLimit > uint256(destChainConfig.maxPerMsgGasLimit)) revert MessageGasLimitTooHigh();
13     if (destChainConfig.enforceOutOfOrder && !evmExtraArgs.allowOutOfOrderExecution) {
14         revert ExtraArgOutOfOrderExecutionMustBeTrue();
15     }
16
17     return evmExtraArgs;
18 }
19
20 /// @dev Convert the extra args bytes into a struct.
21 /// @param extraArgs The extra args bytes.
22 /// @param defaultTxGasLimit default tx gas limit to use in the absence of extra args.
23 /// @return EVMExtraArgs the extra args struct (latest version)

```

```

24 function _parseUnvalidatedEVMExtraArgsFromBytes(
25     bytes calldata extraArgs,
26     uint64 defaultTxGasLimit
27 ) private pure returns (Client.EVMExtraArgsV2 memory) {
28     if (extraArgs.length == 0) {
29         // If extra args are empty, generate default values
30         return Client.EVMExtraArgsV2({gasLimit: defaultTxGasLimit, allowOutOfOrderExecution: false});
31     }
32
33     bytes4 extraArgsTag = bytes4(extraArgs);
34     bytes memory argsData = extraArgs[4:];
35
36     if (extraArgsTag == Client.EVM_EXTRA_ARGS_V2_TAG) {
37         return abi.decode(argsData, (Client.EVMExtraArgsV2));
38     } else if (extraArgsTag == Client.EVM_EXTRA_ARGS_V1_TAG) {
39         // EVMExtraArgsV1 originally included a second boolean (strict) field which has been deprecated.
40         // Clients may still include it but it will be ignored.
41         return Client.EVMExtraArgsV2({gasLimit: abi.decode(argsData, (uint256)), allowOutOfOrderExecution: false});
42     }
43
44     revert InvalidExtraArgsTag();
45 }

```

OffRamp

The following changes have been implemented for any chain support:

- All source chain specific `address` types have been updated to `bytes`
- The `messageId` has been decoupled from `hash(message)` as non EVM chain messages might contain extra fields, different encoding schemes and different hashing algorithms, for example:

Solana OnRamp:

- `messageId = hash(Sol2AnyMessage msg)`
- (Sol2Any message could contain some extra fields that are SOL-specific)

Off-chain:

- The following conversion is then performed off-chain: `Sol2AnyMessage -> Any2AnyMessage -> Any2EVMMessage`

Ethereum OffRamp:

- `committedRootLeafHash = hash(Any2EVMMessage msg)`
- (Any2EVMMessage will contain some EVM-specific fields, but not the SOL specific fields)

`hash(Sol2AnyMessage) != hash(Any2EVMMessage)`

- `metadataHash` has been converted to an implicit hash composed of prefix, source chain selector, destination chain selector and onramp address

```

1 // Hashing all of the message fields ensures that the message being executed is correct and not tampered with.
2 // Including the known OnRamp ensures that the message originates from the correct on ramp version
3 hashedLeaves[i] = Internal._hash(
4     message,
5     keccak256(
6         abi.encode(
7             Internal.ANY_2_EVM_MESSAGE_HASH,
8             message.header.sourceChainSelector,
9             message.header.destChainSelector,
10            keccak256(onRamp)
11        )
12    )
13 );
14
15 /// @dev Used to hash messages for multi-lane family-agnostic OffRamps.
16 /// OnRamp hash(EVM2AnyMessage) != Any2EVMRampMessage.messageId
17 /// OnRamp hash(EVM2AnyMessage) != OffRamp hash(Any2EVMRampMessage)
18 /// @param original OffRamp message to hash
19 /// @param metadataHash Hash preimage to ensure global uniqueness
20 /// @return hashedMessage hashed message as a keccak256
21 function _hash(Any2EVMRampMessage memory original, bytes32 metadataHash) internal pure returns (bytes32) {
22     // Fixed-size message fields are included in nested hash to reduce stack pressure.
23     // This hashing scheme is also used by RMN. If changing it, please notify the RMN maintainers.
24     return keccak256(
25         abi.encode(
26             MerkleMultiProof.LEAF_DOMAIN_SEPARATOR,
27             metadataHash,
28             keccak256(
29                 abi.encode(
30                     original.header.messageId,
31                     original.receiver,
32                     original.header.sequenceNumber,
33                     original.gasLimit,
34                     original.header.nonce
35                 )
36             ),
37             keccak256(original.sender),
38             keccak256(original.data),
39             keccak256(abi.encode(original.tokenAmounts))

```

```

40     )
41   };
42 }

```

- Similarly as in the `FeeQuoter`, `EVM2EVMMessage` structs have been replaced by the `Any2EVMRampMessage` struct.


```

1  /// @notice Family-agnostic header for OnRamp & OffRamp messages.
2  /// The messageId is not expected to match hash(message), since it may originate from another ramp family
3  struct RampMessageHeader {
4    bytes32 messageId; // Unique identifier for the message, generated with the source chain's encoding scheme (i.e. not necessarily abi.encoded)
5    uint64 sourceChainSelector; // —, the chain selector of the source chain, note: not chainId
6    uint64 destChainSelector; // | the chain selector of the destination chain, note: not chainId
7    uint64 sequenceNumber; // | sequence number, not unique across lanes
8    uint64 nonce; // —————/ nonce for this lane for this sender, not unique across senders/lanes
9  }
10
11 struct Any2EVMTokensTransfer {
12   // The source pool EVM address abi.encoded to bytes. This value is trusted as it is obtained through the onRamp. It can be
13   // relied upon by the destination pool to validate the source pool.
14   bytes sourcePoolAddress;
15   address destTokenAddress; // —, Address of destination token
16   uint32 destGasAmount; // —————/ The amount of gas available for the releaseOrMint and transfer calls on the offRamp.
17   // Optional pool data to be transferred to the destination chain. Be default this is capped at
18   // CCIP_LOCK_OR_BURN_V1_RET_BYTES bytes. If more data is required, the TokensTransferFeeConfig.destBytesOverhead
19   // has to be set for the specific token.
20   bytes extraData;
21   uint256 amount; // Amount of tokens.
22 }
23
24 /// @notice Family-agnostic message routed to an OffRamp
25 /// Note: hash(Any2EVMRampMessage) != hash(EVM2AnyRampMessage), hash(Any2EVMRampMessage) != messageId
26 /// due to encoding & parameter differences
27 struct Any2EVMRampMessage {
28   RampMessageHeader header; // Message header
29   bytes sender; // sender address on the source chain
30   bytes data; // arbitrary data payload supplied by the message sender
31   address receiver; // receiver address on the destination chain
32   uint256 gasLimit; // user supplied maximum gas amount available for dest chain execution
33   Any2EVMTokensTransfer[] tokenAmounts; // array of tokens and amounts to transfer
34 }

```

One thing to note is that the `OnRamp` will emit a `EVM2AnyRampMessage` in the `CCIPSendRequested` event while the `OffRamp` `execute()` function will receive `Any2EVMRampMessage` as arguments in its report. The conversion from `EVM2AnyRampMessage` to `Any2EVMRampMessage` is performed off-chain, where the plugins will be in charge of decoding the `extraArgs` from the `EVM2AnyRampMessage` and constructing the `Any2EVMRampMessage`.

Call Flow



1. Dapp or user calls `ccipSend()` on the `Router`.
2. `Router` calls `getFee()` on the `OnRamp` to compute the fee.
3. CCIP v1.6 now outsources the fee specific logic to the `FeeQuoter` so `getFee()` calls `getValidatedFee()` on the `FeeQuoter` which also performs some message validation.
4. `FeeQuoter` will check for reported prices (prices transmitted through the commit plugin or Keystone) staleness and fallback on configured price feeds if stale.
5. `Router.ccipSend()` logic resumes and then calls `forwardFromRouter` on the `OnRamp`.
6. If a message interceptor is set, which for v1.6 will be the aggregate rate limiter, then the hook `onOutboundMessage` function is called on it.
7. `lockOrBurn` is called on the `TokenPool`.
8. The `processMessageArgs()` function is then called on the `FeeQuoter` to convert the message fee to juels, extract extra args variables and perform pool return data validations.
9. If OOO execution is disabled the sender's outbound nonce is incremented on the `NonceManager`.
10. Once the commit DON picks up the `CCIPSendRequested` event it calls `commit()` on the `OffRamp`.
11. If the commit report contains price updated `updatePrices()` is called on the `FeeQuoter`.
12. If the root has already been committed then the transaction reverts.
13. The `OnRamp` then calls `isBlessed()` on the `RMN`. If a root is not blessed then the call reverts, else the root is blessed and the report is committed.

14. Once the execution DON picks up the `CommitReportAccepted` event it calls `execute()` on the `OffRamp`.
15. If a message interceptor is set, which again for v1.6 will be the aggregate rate limiter, then the hook `onInboundMessage` function is called on it.
16. If OOO execution is disabled the sender inbound nonce is incremented on the `NonceManager`.
17. `releaseOrMint()` is called on the `TokenPool`.
18. Finally message is routed to the `Router` for receiver call.

Special Areas of Concern

These are the areas where the auditors should pay special attention:

- DoS: for example a condition that would block all lanes because of a single lane issue
- Security check bypass due to the new multi lane nature of the contracts: for example using the data from one lane to execute a message on another lane
- Things that might have been missed on the chain agnostic side: for example a non EVM chain integration that would somehow break the current security model
- Hashing collisions
- Replay attacks messages/price reports (same chain/cross chain)
- Upgrades leading to unintentional side effects
- Nonce ordering not honoured
- Nonce ordering leading to stuck messages
- Multi-source batch getting stuck (one source blocking another)
- Unintentional interactions between 1.5 self-serve pools and V2 multi-ramps
- Over/under payment / billing bugs
- Wasting NOP gas
- Unrecoverable states in the CCIPHome / RMNHome state transitions

Known / Out of Scope Issues:

Token developer is malicious

- Any malicious token developer can negatively impact transactions that contain their token.
- Any malicious token developer can make transactions that contain their token fail forever, leading to the loss of any funds in that transaction.
- Users should be cautious about sending tokens they do not trust.
- Sending a single token per transaction will prevent any contagion to other tokens.
- On some chains, there are methods through which a malicious token developer can cause an entire transaction to revert without being caught by the try-catch, such as overflowing ZK proof size on EVM-compatible zkRollups. In such edge cases, a malicious token developer can block following transactions from the same sender at the destination when not using out-of-order execution, potentially causing unbounded loss.
- Anyone can add any token with any name or affiliation. The CCIP Owner has zero control over what is listed and cannot modify or remove any tokens. This means there could be tokens listed with names or affiliations that are not endorsed by the CCIP Owner.
- Malicious tokens or token pools will be able to re-enter.
 - This should not have any negative impact on the CCIP system.
 - This does allow them to re-order some events, but never to double spend.

Token developer responsibility

- When a token developer misconfigures their own tokens/pools, users could potentially not receive these tokens.
- The CCIP Owner has no special access to resolve misconfigurations, only the token developer can do so.
- Token developers can deploy token pools without rate limits, which is different from today where every token pool is rate-limited.

Compatibility

- Some tokens will not work.
- For example, fee-on-transfer tokens won't work because they require multiple hops in the CCIP contracts.
- They will still be able to register their token, as this process is fully permissionless. This could negatively impact transactions containing these tokens.
- Pool `releaseOrMint` is bound by a per-ramp `maxPoolReleaseOrMintGas`, and Token transfer is bound by a per-ramp `maxTokenTransferGas`. As a result, pools or tokens that are extremely gas-intensive may not be supported. Similarly, pool data that can be relayed from source to destination is bound as well.
- Not every token will be able to permissionlessly register such as tokens that do not expose the `owner` and don't have the `getCCIPAdmin` function.
- These can still be onboarded in the same way all current tokens are onboarded: manually through the CCIP Owner.

Other

- CCIP Owner is a trusted role.
- The multi-signature contract structure includes the entire DON and therefore inherits the security of the entire DON.
- Solidity 0.8.24 is used and CCIP will be deployed on various chains that don't support the newer Solidity features.
 - The Paris hard fork is explicitly compiled to ensure compatibility.
- The aggregate rate limiter only works for tokens that have prices, as it's denominated in USD.
- Self-serve tokens will not have prices, as it would be impractical to acquire accurate price sources for any arbitrary token, and writing these prices onchain would be economically unsustainable, and easily exploitable.
- This means that tokens default to *not* be included in the aggregate rate limiter.
- Router API `getSupportedTokens()` is now deprecated, calling it will revert.

- There is no longer an official API that returns all supported tokens for a given destination in 1 call.
- One should iterate through tokens via `getAllConfiguredTokens()` in `TokenAdminRegistry` and then call `isSupportedChain()` on its token pool.